

## A Handy use of the %LINE Annotate Macro

Deena Rorie, University of Arkansas, Fayetteville, AR  
Lynette Duncan, University of Arkansas, Fayetteville, AR

### ABSTRACT

SAS/GRAPH<sup>®</sup> is a powerful and flexible tool for generating plots. The Annotate facility provides even more flexibility. Most new SAS/GRAPH programmers shy away from the Annotate facility, but the macros available in the Annotate facility are simple to use and should be a part of every SAS/GRAPH programmer's toolbox. This paper demonstrates the usefulness of the %LINE macro in a GPLOT graph and is directed to users of SAS/GRAPH who may or may not have been brave enough to use the Annotate facility.

### INTRODUCTION

In the Center for Statistical Consulting, we sometimes receive requests for plots that our clients cannot produce with their usual software. These plots often come to us as hand-sketches, and our job is to draw these plots using SAS. Over the past few years, SAS has made great strides in improving the GUI method of producing plots, but for atypical plots, we rely on the power of SAS/GRAPH and, in particular, the Annotate facility. We demonstrate the use of the Annotate %LINE macro in a GPLOT graph.

### REQUEST

We received a hand drawn sketch for a plot that contained 3 y-variables "stacked" on the y-axis with dotted grid lines and solid vertical reference lines. (See Figure 1 for the end product.) Our first reaction was, no problem! (Thinking to ourselves that we could use the GRID and VREF options.)

We quickly discovered that if our grid lines were dotted, our vertical reference lines would also be dotted. If we tried to make the vertical reference lines solid, the gridlines became solid. Our first attempt is shown in Figure 2.

We began to ask ourselves: "How can we draw a line in a plot?" After contemplating the question for a while, we finally remembered the %LINE macro in the Annotate facility.

### BASICS OF THE ANNOTATE FACILITY

To use the Annotate facility with GPLOT, two things are needed: a SAS dataset containing variables telling the Annotate facility what to draw and the ANNOTATE=*annotate-dataset* option in the PLOT statement.

The dataset is an ordinary SAS dataset, but the names of the variables make it a dataset that can be used with the Annotate facility.

### REQUIRED VARIABLES

In order to use the %LINE macro, three variables must be implicitly created in the annotate dataset.

- **HSYS** defines the coordinate system used by the SIZE variable. The value of the variable is a number between 1 and 9 or A, B, or C. We use 2, which specifies that the data values define the coordinate system.
- **YSYS** and **XSYS** define the coordinate system used by Y and X variables respectively. The value of these variables is a number between 1 and 9 or A, B, or C. We use 2, which specifies that the data values define the coordinate system.

### %LINE MACRO

The %LINE macro is a built-in annotate macro, and the %ANNOMAC statement is necessary to make this macro available. The %ANNOMAC statement and the %LINE macro

are both included in the data step used to create the annotate dataset.

The %ANNOMAC statement appears in the data step just before the %LINE macro. (See the SAS Code below). The %LINE macro draws a line between any two points on the plot, ( $x_1, y_1$ ) and ( $x_2, y_2$ ). The arguments for the %LINE macro are as follows:  $x_1, y_1, x_2, y_2$ , line color, line type, and width of line. Based on these arguments, the following variables are created automatically:

1. X and Y. The values of these variables are the x and y coordinates of the endpoints of the line.
2. COLOR. This variable defines the color of the line. Any valid SAS color may be used.
3. LINE. This variable determines the line type (solid or dashed) of the line. The value 1 produces a solid line, and the values 2 through 46 produce dashed and dotted lines.
4. SIZE. The SIZE variable determines the thickness of the line.
5. FUNCTION. When used with the %LINE macro, this variable has the value MOVE or DRAW. The value tells the Annotate facility what to do and where to do it.

### THE ANNOTATE DATASET

Following is the SAS code for the data step used to create the Annotate dataset. Notice that the %LINE macro is called twice because we need to draw two lines. (The name of the dataset need not be ANNOTATE.) The data are shown in Table 1.

```
DATA ANNOTATE;
RETAIN HSYS YSYS XSYS '2';
%ANNOMAC;
%LINE ('9:40't,2000,'15:10't,2000,BLACK,1,.1);
%LINE ('9:40't,1000,'15:10't,1000,BLACK,1,.1);
RUN;
```

Table 1. The dataset generated by the data step.

Obs	HSYS	YSYS	XSYS	color	X
1	2	2	2	BLACK	34800
2	2	2	2	BLACK	54600
3	2	2	2	BLACK	34800
4	2	2	2	BLACK	54600

Obs	Y	FUNCTION	LINE	SIZE
1	2000	MOVE	.	.
2	2000	DRAW	1	0.1
3	1000	MOVE	1	0.1
4	1000	DRAW	1	0.1

The  $y_1$  and  $y_2$  arguments may look incorrect when compared to the plot in Figure 1. However, 1000 and 2000 are the actual locations of lines on the y-axis. The VALUE= option was used in the AXIS statement to label the y-axis. (The complete SAS code appears in the next section.)

The variables HSYS, YSYS, and XSYS were specified in the data step, but the other variables were generated automatically in the call to the macro. Each call to the %LINE macro generated 2 observations. The first observation contains the MOVE function that directs GPLOT to move its pointer to the ( $x_1, y_1$ ) coordinate. The DRAW function tells it to draw the line from its current location, ( $x_1, y_1$ ), to the final location, ( $x_2, y_2$ ).

### CREATING THE PLOT

Generating the Annotate data set will not affect the plot unless it is included in the PLOT statement of PROC GPLOT. The SAS Code shown below

1. Rearranges the data for plotting
2. Specifies axis, symbol, and legend options
3. Generates the plot with the annotate dataset.

#### STEP 1: REARRANGE THE DATA FOR PLOTTING

The original data contained variables S0, S1, V1, and time. S0 and S1 were on the same scale, and V1 was on a larger scale. In order to "stack" the three y variables, their scales were changed.

```
DATA PLOT;
  SET TRAFFIC;
  Y=S0*10;      Z='S0';  OUTPUT;
  Y=S1*10+1000; Z='S1';  OUTPUT;
  Y=V1+2000;   Z='V1';  OUTPUT;
  KEEP STARTTIME Y Z;
RUN;
```

#### STEP 2: SPECIFY AXIS, SYMBOL, AND LEGEND OPTIONS

The AXIS1 statement was used to modify the y-axis so that the tick mark labels were correct. The ORDER= and VALUE= option controlled which major tick marks appear and how they were labeled.

The AXIS2 statement controlled options on the x-axis. The SYMBOL statements changed the size and color for the three variables on the y-axis. The LEGEND statement controlled the label and font size of the legend.

```
AXIS1 LABEL=(A=90 H=10PT 'S0  S1 V1')
      ORDER=(0 TO 5000 BY 250)
      VALUE=(H=10PT '0' '25' '50' '75' '0' '25'
'50' '75' '0' ' ' ' ' ' ' ' ' ' ' ' ' ' ' ' ' ' '
'2000' ' ' ' ' ' ' ' ' ' ' ' ' ' ' ' ' ' ' ' '
'3000');
AXIS2 ORDER=('9:40't TO '15:10't BY 1200)
      OFFSET=(,0.1875IN)
      VALUE=(H=10PT A=90) MINOR=(NUMBER=1)
      LABEL=(H=10PT);
SYMBOL1 H=.25 V=DOT C=BLUE;
SYMBOL2 H=.25 V=DOT C=GREEN;
SYMBOL3 H=.25 V=DOT C=RED;

LEGEND1 LABEL=NONE VALUE=(H=10PT);
```

#### STEP 3: GENERATE THE PLOT WITH THE ANNOTATE DATASET

The GPLOT statement was of the form  $Y*X=Z$  so that there were

different symbols for each level of Z (this is the variable that defined whether Y was S0, S1, or V1). The VAXIS, HAXIS, and LEGEND options refer to the statements in Step 2. The GRID option drew gridlines at each major tick mark, and the ANNOTATE= option referred to the Annotate dataset. Again, the name of this data set in our program was ANNOTATE, but it could have been anything.

```
PROC GPLOT DATA=PLOT;
  PLOT Y*STARTTIME=Z/VAXIS=AXIS1 HAXIS=AXIS2
  LEGEND=LEGEND1 GRID ANNOTATE=ANNOTATE;
  TITLE1 H=10PT 'Plot with the Annotate Macro';
  LABEL STARTTIME='Time';
RUN;
QUIT;
```

#### CONCLUSION

SAS/GRAPH is a flexible tool for generating graphs in SAS. The Annotate facility in SAS/GRAPH provides the ability to enhance graphs beyond what is available in the procedures. The Annotate facility can be difficult to learn, but the built-in macros are the easiest part of the facility to use. Even a beginning user of SAS/GRAPH could take advantage of these macros. The %LINE macro, in particular, is straightforward.

#### REFERENCES

SAS Institute, Inc. (2000), *SAS OnlineDoc Version 8*, Cary, NC: SAS Institute, Inc.

#### CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

University of Arkansas  
 Department of Mathematical Sciences  
 SCEN 301  
 Fayetteville, AR 72701  
 (479) 575-6429  
 duncan@uark.edu

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.




Figure 1. Final graph.




Figure 2. First attempt.